

## NAME

spicemod - model file for SPICE

## DESCRIPTION

The file *spicemod* contains model information for the SPICE simulator. This file is used by *xspice* to prepare simulation input. However, the *spicemod* file can also be used by *xspectre* for the Spectre simulator.

The *spicemod* model file has a very simple format:

The lines above the '%' line contain a model index. These lines are used to specify device cell names. Only models specified in this index can possibly be included in the output. The lines can also be used to specify alternate output names and (if appropriate) correction values for device "w" and "l" attributes. The first argument on a line specifies an alternate output name. The second and third arguments are the "dw" and "dl" values. Note that the output name must be used in one of the specified models below the '%' line, else no model type is known or it must be a predefined model. And also no model lines can be included in the output.

The lines below the '%' line contain the models and possibly lines to specify "nbulk" and "pbulk" voltages and resistors. Only the used *model* lines after the '%' line are copied to the *xspice* output file. The comment lines and all other lines are not copied. The keywords *vnbulk* and *vpbulk* can be used to set the "nbulk" and "pbulk" voltages. The keywords *rnbulk* and *rpbulk* can be used to set the "nbulk" and "pbulk" resistor values. They are *xspice* control lines, which are used but not outputted. Thus the specified node numbers or names are not used.

Some remarks:

Do not use other separator characters than space and tab. Place ".model", *model\_name*, *model\_type* and the "level=N" parameter all on the same line. Use a lowercase modeltype. When the *level* parameter is used, it must be the first ".model" parameter. The *xspectre* program uses this for doing model conversion for the Spectre simulator. If needed, the SPICE scale factors are also converted. But scale factors 'a' and 'mil' are not supported.

The example below:

The lines above the '%' line contain the *xspice* or *xspectre* control information. The '#' line is a comment line. The "nenh" device must get the output name "nenhs" and every "w" must be corrected with a "dw" of "0.88" micrometer. The "l" attribute must not be corrected ("dl" is 0). Note that a *positive* value gives a *smaller* device attribute. Note that "nenh" and "ndep" are predefined device names, thus the used model-type of the models must be "nmos".

## EXAMPLE

For example, this is the *spicemod* file of the NMOS process:

```
# rcsid = "spicemod 1.2 2003/05/02 07:56:32 simon"
nenh nenhs 0.88e-6 0
ndep deps 1.20e-6 0
%

* parameters for depletion NMOS w/l = 4/4
.model deps nmos level=2
+   vto=-3.971 kp=31.54u gamma=0.3642
+   phi=0.578 pb=0.84 cgso=430p
+   cgdo=430p cgbo=130p rsh=25.3
+   cj=62u mj=0.73 cjsw=340p
+   mjsw=0.20 js=6.2u tox=0.07u
+   xj=0.35u ld=0.605u uo=640
+   ucrit=534000 uexp=-0.025 utra=0
+   xqc=0.5 fc=0.6 lambda=0.0258
+   vmax=3.5e4 nsub=9.8e14 delta=6.3
```

```

* end deps w/l = 4/4

* parameters for n_enhancement NMOS w/l = 4/4
.model nenhs nmos level=2
+      vto=0.6838  kp=32u    gamma=0.2787
+      phi=0.55   pb=0.84   cgso=440p
+      cgdo=440p  cgbo=100p  rsh=25.3
+      cj=62u     mj=0.73   cjsw=340p
+      mjsw =0.20  js=6.2u   tox=0.07u
+      xj=0.35u   ld=0.625u  uo=650
+      ucrit=143000 uexp=0.154  utra=0
+      xqc=0.5    fc=0.6    lambda=0.02269
+      vmax=6e4   nsub=5.71e14 delta=5.5
* end nenh w/l = 4/4

vnbulk 1 0 -2.5V
rnbulk 1 0 100meg

* vdd ? 0 5V
* rgnd ? 0 1p
* rvss ? 0 1p

```

#### **AUTHOR**

S. de Graaf

#### **SEE ALSO**

SPICE2 / SPICE3 User's Guide,  
xspectre(1ICD), xspice(1ICD).